

Lecture 15:

Register Transfer Language and Verilog

COS / ELE 375

Computer Architecture and Organization

Princeton University
Fall 2015

Bochao Wang

(Based on slides by David Penry and Neil Vachharajani)
(Prof. David August)

1

RTL Tools

RTL, just like (most) programming languages:

- Precise and unambiguous
- Programmer must debug
- Code can be checked automatically for certain properties
 - Type checking
 - Check RTL model against more abstract state machine
- Tools can process the code
 - RTL → simulator
 - RTL → processor
 - RTL →



3

RTL (Register Transfer Language)

- Designing processors at the gate level is difficult
- Use a higher-level language

RTL: a language for describing the behavior of computers in terms of step-wise register contents

2

Hardware Description Languages

- Used to describe hardware for simulation and synthesis
- Why? Designs are too big to just draw schematics
- Major languages:
 - Verilog
 - Cadence, Inc.
 - Emphasis on practical use (I/O well-defined, ability to record)
 - Possibly most widely used
 - VHDL
 - Department of Defence
 - Emphasis on abstraction (derived from Ada)
 - Popular in academic circles, Europe; required of defence contractors
 - SystemC
 - Synopsys, Inc.
 - Emphasis is "higher-level" simulation – really just a C++ template library
 - Up-and-coming

4

Simulation vs. Synthesis



- HDLs are used for both simulation and synthesis
- Simulation: does the design work?
 - Can also be used for "does the design meet timing?"
- Synthesis: generate a circuit that is equivalent
 - Recognize state and combinational logic
 - Optimize the circuit

Not all valid HDL programs are synthesizable

5

Example


Modules have initial, **continuous**, and always blocks



```
module example(a, b, c, f, e);
  input a, b, c;
  output f, e;
  wire d;
  and g1(d, a, b);
  not g2(e, c);
  or g3(f, d, e);
endmodule
```


Youpyo Hong, Dongguk University

Verilog Design Flow



6

Modified Example



```
module example(a, b, c, f, e);
  input a, b, c;
  output f, e;
  assign d = a & b;
  assign e = ~c;
  assign f = d | e;
endmodule
```

Youpyo Hong, Dongguk University

- Use **assign** statement for combinational circuit
- Note the bit-wise operators (implies wire width...)

7

8

- ; is the statement terminator (terminator, not separator)
- /* */ for multi-line comments
- // can be used for single-line comments
- Constants can have bit-width and radix:
<bit-width>'<radix><val>
32'h1234ab34 = 1234ab34₁₆
8'b0100_0110 = 01000110₂; note that underlines are legal

Elements of Verilog

9

Datatypes

- 4-valued logic: 0, 1, x, z
 - 0, 1 – normal binary 0 and 1
 - z – high-impedance value: this is what tri-state buffers drive when they are off
 - x – simulator doesn't know; sometimes used as "don't care"
- Integers – don't use these outside of test benches
- At simulation start, all signals have x's in every bit
- Unconnected inputs to a module have value 'z'.

10

Module Overall Structure

```
module the_design ( input, output );
  Declarations: ports, constants, variables(wire, reg)
  Instantiations of other modules
  Continuous assignment: assign y = ...
  Behavioral statements (initial, always) {
    procedural blocking assignment
    procedural nonblocking assignment
  }
endmodule
```

11

12

Signals

- Signal: a “variable” that can have values assigned “ahead of time”
- Can be thought of as a wire or a group of wires
- Syntax:
<kind> [<width specifier>] name, name, ...;
- Examples:

```
reg [15:0] IR;
wire [3:0] a; a[3] is MSB
wire [1:8] b; b[1] is MSB //not recommend
wire doit;
```
- Registers vs. wires:
 - Reg: inside “always” block, lhs, e.g. IR = ...
 - Wire: outside “always” block, lhs, e.g. assign a = ...

13

Register Transfer Language?

- Variables correspond to the hardware registers
- “always” blocks

```
Module D_FlipFlop(Q, D, CLK);
    output Q;
    reg Q;
    input D, CLK;
    always @ (negedge CLK) // Sensitivity List
    begin // Not needed for 1 statement
        Q <= D; // Note: No assign
    end
endmodule
```

15

Wires

- Wires need not be declared unless it is multiple wires

```
module example(b, c);
    input b;
    output c;
    wire a;           // not necessary
    wire [3:0] d;    // necessary
    ...
    assign c = b | a;
endmodule
```

14

Module Overall Structure

```
module the_design ( input, output );
    Declarations: ports, constants, variables(wire, reg)
    Instantiations of other modules
    Continuous assignment: assign y = ...
    Processes(initial, always) {
        procedural blocking assignment
        procedural nonblocking assignment
    }
endmodule
```

16

Modules

- Allow you to organize the design hierarchically
- Allow structural reuse of the design

- Defining a module:

- Example:

```
module flop (clk, D, Q); // name, list of ports
    input clk;           // define directions of ports
    input D;
    output Q;
    ...
    // code to do things
endmodule
```

- Instantiating a module:

```
flop U1 (.clk(someclk), .D(someD), .Q(someQ));
```

17

Module Overall Structure

```
module the_design ( input, output );
    Declarations: ports, constants, variables(wire, reg)
    Instantiations of other modules
    Continuous assignment: assign y = ...
    Processes(initial, always) {
        procedural blocking assignment
        procedural nonblocking assignment
    }
endmodule
```

18

Kinds of assignments


- Continuous (`assign =`)
 - Only to "wire"
 - Always sensitive to things on the right-hand side
- Blocking (`=`)
 - Made inside a process
 - Only to "reg"
 - Value of lhs changes immediately
- Non-blocking (`<=`)
 - Made inside a process
 - Only to "reg"
 - Value of lhs changes only after all rhs have been evaluated
 - A time-spec after the statement says to wait before changing the value
 - Continue executing process even if there is a time-spec

Assignment

<code>a = b</code>	Do Sequentially (blocking, sequentially)
<code>A <= b</code>	Do RHS first (nonblocking, grab at t=0)

Which 3 are Shift Registers?

```
Z<=Y; Y<=X;
Y<=X; Z<=Y;
Z=Y; Y=X;
Y=X; Z=Y;
```



19

20

Processes

- A group of sequential statements; three kinds:
 - Initial - runs only at simulation start
 - Always – runs over and over
 - Continuous – special syntax for a common kind of always block
- Can be suspended, waiting for something to happen
 - For a signal in a group of signals to change
 - For an amount of time to pass
 - For a signal to become true (I've never needed this)

21

Assignment examples

```
reg rst_1, Q;
wire cachehit;

initial begin
    rst_1 = 0;          // blocking
    #300;
    rst_1 = 1;          // blocking
end

always @(posedge clk)
    Q <= #1 D; // non-blocking

assign cachehit = comparator_matches &
    valid; // continuous
```

23

Process examples

```
reg rst_1, Q;
wire cachehit;

initial begin
    rst_1 = 0;
    #300;
    rst_1 = 1;
end

always @(posedge clk)
    Q <= #1 D;

assign cachehit = comparator_matches & valid;
```

22

Blocking a process

- Wait for signals to change:
`always @(signall or signal2)`
- Wait for a signal edge:
`always @(posedge clk or negedge rst_1)`
- Wait for time:
`# 300;`
`# 300 a = b; // can also put before a statement`
- Wait for logical value:
`wait(b); // I've never used this`

24

Operators

- Most of the operators look like C:
 - Bitwise operators: &, |, ~, ^
 - Logical operators: &&, ||, !
 - Comparisons: ==, !=, <, <=, >, >=
 - If either operand is x or z, they return FALSE.
 - Arithmetic: +, -, *, /, %, <<, >>
 - Choice: ?:
- Order of operations may differ from C; use parenthesis
- Special operators:
 - Comparison with x or z values: ===, !==
 - Bitwise operators can be used as reduction operators:
 $(| x)$ = or all the bits of x together.

25

Case statements

- Rather different from C switch statements:

```
case (myvar)
  1 : dothis = 1;
  2 : dothat = 3;
  3 : dosomemore = 4;
default: noneoftheabove = 1;
endcase
```
- Note that there are no fall-throughs
- Multiple statements inside one case require `begin/end` around them

27

Statements

- Control structures
 - C-like: `if`, `while`, `for`
 - `forever` – do following statement forever
 - `repeat (<#>)` – do following statement a number of times
 - `case` (see next slide)
- Parallel control structures
 - `fork/join` – do statements in between in parallel
- Signal overrides
 - `force` – override the value of a signal
 - `release` – stop overriding the value of a signal

26

Example: 4 to 1 Mux

```
module mux4_to_1 (out, i0, i1, i2, i3, s1, s0);
  output out;
  input i0, i1, i2, i3;
  input s1, s0;
  reg out;
  always @(s1 or s0 or i0 or i1 or i2 or i3)
    begin
      case ({s1, s0})
        2'b00: out = i0;
        2'b01: out = i1;
        2'b10: out = i2;
        2'b11: out = i3;
        default: out = 1'bX; // X is don't care
      endcase
    end
endmodule
```

28

Parameters

- Increases ability to reuse modules
- Defining a parameter (with a default value):

```
parameter me = 3;
```

- Overriding the default value

```
defparam myunit.me = 4;
```

- Can also use a #(value) syntax when instantiating for modules with small numbers of parameters:
e.g. module my_module #(parameter BIT_WIDTH = 32, ...)
(input clk, input [BIT_WIDTH-1:0] addr, output a...)

29

Preprocessor directives

- Always begin with a back-tick
- ``define name value`
 - defined values are referenced as ``name`

Using the mux module

- It's a 4-input mux for busses with a 'width' parameter for the size of the bus

```
wire [7:0] A,B,C,D, result;  
wire [1:0] sel;
```

```
mux4 mymux (.out(result), .in0(A), .in1(B),  
            .in2(C), .in3(D), .sel(sel));  
defparam mymux.width = 8;
```

OR

```
mux4 #(8) mymux (.out(result), .in0(A), .in1(B),  
                  .in2(C), .in3(D), .sel(sel));
```

30

A Sample Finite State Machine

```
reg [1:0] state;  
`define IDLE 2'b00  
`define READ 2'b01  
`define WRITE 2'b10  
`define DONE 2'b11  
  
always @ (posedge clk or negedge rst_1)  
  if (~rst_1) state <= `IDLE;  
  else  
    case (state)  
      `IDLE:  
        if (req)  
          if (wr) state <= `WRITE;  
          else state <= `READ;  
        else state <= `IDLE; // not strictly necessary  
      `READ:  
        state <= `DONE;  
      `WRITE:  
        state <= `DONE;  
      `DONE:  
        state <= `IDLE;  
      default:  
        state <= `IDLE;  
    endcase
```

31

32

Dealing with Synthesis

- Things you can't synthesize reliably:
 - Some for loops (hard to guess)
 - @() in the middle of a process
 - **Time specifiers**
 - force/release
- Watch out for hidden latches!

```
always @(a)
  c = a | b;
```

 - Actually creates a latch
 - All things on the rhs must go on the sensitivity list unless you are trying to create a state element
- Using don't cares:
 - If you really don't care about a value, assign x to it, and the synthesis tool will choose a value which simplifies the logic

33

My Rules

- Make all flops (edge-sensitive always blocks) use a non-blocking assignment with #1 delay
 - Easier to see what's going on in waveforms
- Make all other always blocks use blocking assignment and use complete sensitivity lists
 - always @(*) for next state logic combinational
 - always @ (posedge clk) for sequential
- Do combinational logic in continuous assignments except for next state logic

34

Another Sample Finite State Machine

```
module fsm
(
  input  clk,
  input  rst,
  input  in,
  output redOut,
  output yellowOut,
  output greenOut
);

localparam RED = 0;
localparam YELLOW = 1;
localparam GREEN = 2;

reg [1:0] state_f;
reg [1:0] state_next;

// Combinational Logic
always @ *
begin
  state_next = state_f;
  case (state_f)
    RED:
      if (in)
        state_next = GREEN;
    YELLOW:
      if (in)
        state_next = RED;
    GREEN:
      if (in)
        state_next = YELLOW;
      default: state_next = RED;
  endcase
end

// Sequential Logic
always @ (posedge clk)
begin
  if (rst)
    state_f <= RED;
  else
    state_f <= state_next;
end

  assign redOut = (state_f == RED);
  assign yellowOut = (state_f == YELLOW);
  assign greenOut = (state_f == GREEN);
endmodule
```

35

Thank you

36

Using the regfile2 module

- You should use this module for your register file.
 - It synthesizes properly
 - It has a nice task for displaying the register file contents
- Interesting characteristics:
 - One read ports, one read/write port
 - Port A writes if the enable bits are set (bytes are controlled individually); it will always read as well
 - Port B can only read
 - Clocked on the positive edge of the clock (you can't do the write in first half, read in second half trick from P & H)
- If you name your datapath DATA (the instance name, not the module name), and make IR_Enable a signal indicating when the IR is to be loaded with a new instruction, you can uncomment code in monitors.v to get automatic register dumps before each instruction.

37

I/O

- Displaying data:
`$display(<format string>, ...);`
 - Like printf, but
 - %b gives binary; %h gives hex
 - %t formats a time value
 - automatically puts on a newline
- Recording signals:
 - `$dumpfile(<file name>);` – set the dumpfile name
 - `$dumpvars(<# levels>, <hierarchy>);`
 - Adds signals below a particular point in the hierarchy to the list of signals to dump; only decends the hierarchy for the number of levels specified; 0 means no limit
 - `$dumpon;` - turns on dumping
 - `$dumpoff;` - turns off dumping

39

Functions and tasks

- Functions are like C functions, but cannot have side effects
- Tasks do not have return values, but can have side effects

- Examples:

```
function [31:0] add1;
    input [31:0] x;
begin
    add1 = x + 1;
end
endfunction

task addtask;
    input [31:0] x;
    output [31:0] y;
begin
    y = x + 1;
    mycrazysignal = 3;
end
endtask
```

38

Simulation control

- `$finish;` – end simulation and exit the simulator
- `$stop;` – stop simulating; go to the command-line interface

40